

# SAS Technical News

Spring 2002

*For Higher  
Customer Satisfaction,  
We Bridge  
the SAS System  
Between  
Customer's World.*

## CONTENTS

- 1 特集 SASシステムで実現できるセキュリティ設定

---

- 10 Q&A

---

- 17 新刊マニュアルのご紹介

---

- 18 SASトレーニングのお知らせ

---

- 19 最新リリース情報

---

- 19 SASプログラマーズ・ワークショップ開催のお知らせ

## 特集

# SASシステムで実現できる セキュリティ設定

### はじめに

昨今、インターネットを含むネットワーク環境の複雑化・肥大化・大衆化に伴い、ネットワークサーバー等への不正侵入などセキュリティに関する問題が重要視されるようになりました。ネットワークを使つての利用が可能なSASシステムにおいても、セキュリティ対策について無視できない問題となっています。今回の特集では、従来から提供されていた、SASデータセットに対するセキュリティやネットワークに関するセキュリティについてご紹介します。

1. SASデータセット等に対して設定するセキュリティ  
SASシステムで生成されるさまざまなファイル(データセット、カタログ等)は、他のユーザーによる参照・改変の危険性があります。これらのファイルを保護するための手法について、いくつかご説明します。

#### 1.1 OSの機能を利用したSASデータセット・カタログに対する権限付与

SASデータセットおよびカタログについて、参照や改変の権限を制御する最も簡単な方法は、ディレクトリやフォルダごとにオペレーティングシステム(OS)側で適切な権限を付与することです。SASシステムを管理者権限のユーザーでインストールした後、SASシステムを通常利用する範囲では、各ユーザーに以下の権限が最低限必要となります。

権限	権限を制限できるファイル・ライブラリ
参照権限のみ	SASシステム本体(モジュール群)
参照権限または全権限(参照・更新)	SASUSERライブラリ 原則として、SASUSERライブラリには全権限が必要です。参照権限のみで使用するには「RSASUSER」オプションを指定してください。ただしSASUSERライブラリ内に格納されるプロファイル等の各種設定作業が完了している必要があります。
全権限(参照・更新)	WORKライブラリ(処理時に作成される中間ファイル等が格納されるため、必要です) その他ユーザーが利用するライブラリ

## 1.2 SASシステム固有のSASデータセットに対するセキュリティ機能

SASデータセットに対するセキュリティ機能として、「パスワードデータセットオプション」「ENCRYPTデータセットオプション(暗号化)」が提供されています。

### パスワード機能

パスワード設定用として用意されている「READ=」「ALTER=」「WRITE=」「PW=」データセットオプションを適切に設定することで、各処理に対するセキュリティを実現できます。

ALTER=	データセット構造の変更・削除に対するパスワード設定
READ=	データセット構造やオブザベーションの参照に対するパスワード設定
WRITE=	オブザベーションの追加・変更・削除に対するパスワード設定
PW=	変更・参照・作成の全処理に対するパスワード設定

データセット構造の変更とは、変数名変更や、インデックス作成・削除などの操作を意味します。

### 設定プログラム例

データセットpwdtestに全処理用のパスワードを設定する場合

```
LIBNAME pwdlib 'd:\temp' ;
/* 全処理に対するパスワード設定 */
DATA pwdlib.pwdtest(PW=pwdtest) ;
  LENGTH A $ 20 ;
  A = 'パスワードテスト1' ;
  OUTPUT ;
  A = 'パスワードテスト2' ;
  OUTPUT ;
  A = 'パスワードテスト3' ;
  OUTPUT ;
RUN ;

/* 参照処理のパスワード設定 */
PROC PRINT DATA=pwdlib.pwdtest(READ=pwdtest) ;
RUN ;
```

上記プログラムでパスワード設定されたSASデータセットに対しては、一部の例外を除き以下の作業も行なえます。

- ・「SAS System Viewer」や「Enterprise Guide」を使つての参照パスワード入力用のウィンドウが表示されます。
- ・「SAS ODBC Driver」によるMS-AccessからのSASデータセットのインポート  
「オブジェクト'xxxx.xxx'が見つかりませんでした。オブジェクトが存在していること、名前やパス名が正しいことを確認してください。」のメッセージが表示され、SASデータセットの認識がされません。一度テーブルとして読み込むと変更は可能となりますので、変更および作成用のパスワードについては全く意味がなくなります。

### 暗号化

参照用のパスワードを設定したSASデータセットであっても、ファイル自体をバイナリーエディタ等で表示させれば、データの中身はある程度閲覧できてしまいます。右記は、バイナリーエディタを使用してデータセットを表示させた例です。



このような方法に対して有効な手段が、暗号化です。暗号化は下記のようにENCRYPT=データセットオプションにより行ないます。

```
ENCRYPT = YES | NO
```

デフォルトはNO

このENCRYPT=データセットオプションを設定する場合、同時に「PW=」あるいは「READ=」データセットオプションを設定する必要があります。

設定プログラム例)

```
/* 暗号化と参照用パスワード設定 */
DATA pwdlib.pwdenc (ENCRYPT=yes READ=test) ;

  LENGTH A $ 20 ;
  A = '暗号化テスト1' ;
  OUTPUT ;
  A = '暗号化テスト2' ;
  OUTPUT ;
  A = '暗号化テスト3' ;
  OUTPUT ;
RUN ;
```

暗号化されたデータセットをバイナリーエディタ等で表示させると、以下の表示になり、内容を閲覧できなくなります。



なお、暗号化されたデータセットを読み込む場合、暗号データの復元処理が発生するため、パフォーマンスが劣化します。読み込みの速度はCOMPRESSデータセットオプションにより圧縮されたデータセットの読み込みとほぼ同等です。

## 1.3 アクセスコントロールを使ったデータ・アプリケーションに対するセキュリティ

SASシステム バージョン8以降、アクセスコントロール機能が実装されました。この機能はSAS/EISソフトウェアで提供されており、この機能で可能なメタベース(テーブル、多次元データベース)、EISアプリケーション、関数に対する使用権限を細かく設定することができます。この機能を利用できるのは、SAS/EISソフトウェアおよびそれと連携できる製品に限られますが、ここで簡単に触れておきます。(詳細な説明は「SAS OnlineDoc. V8」をご参照ください。)

### (1) アクセスコントロール機能で実現できること

SAS/EISソフトウェア起動時に表示される「Login」画面で入力するユーザーID(正確にはそのユーザーIDが所属するグループ)に応じて、「メタベース」「EISアプリケーション」「関数」のうちで使用できるものを制限することができます。具体的に設定できる項目は右記のものです。

メタベース	メタベース自体の非表示、特定列のみの保存および削除、列の非表示、階層初期表示の位置指定など
EISアプリケーション	特定のアプリケーションのみの表示、特定アプリケーションの実行不可(表示はされる)
関数	特定機能の実行不可(「メタベースの編集」「アプリケーションの作成」などの機能)

(2)アクセスコントロールを利用可能にするには紙面の都合上、詳細な説明は割愛しますが、アクセスコントロール機能を使用するためには以下の作業が必要です。

- ①メニューの「ソリューション」「EIS/OLAPアプリケーションビルダ」にてSAS/EISソフトウェアを起動します。「EISメインメニュー」にある「設定」の「多次元アプリケーション」「アクセスコントロール」を選択して、「アクセス制御設定ウィンドウ」を起動します。
- ②「アクセス制御設定ウィンドウ」の「アクセス制御キー」をクリックし、適当なキーを設定してください。以後、このキーを知っている者のみが「アクセス制御設定ウィンドウ」を起動できます。キー設定後、SAS/EISソフトウェアを再起動してください。
- ③「キー入力ウィンドウ」で②において設定した「アクセス制御キー」を入力し、再び「アクセス制御設定ウィンドウ」を開きます。そして「環境設定」の設定を行ないます。

ACLルートパス	ユーザーID/パスワード/アクセスコントロール情報を格納するパス
ホスト	他のマシンのSAS/EISソフトウェアに対して設定する場合、変更の必要がある
ログインウィンドウ	ユーザーID入力画面のタイプを変更する

上記②～③の設定終了後の『アクセスウィンドウ』の状態例



設定終了後SAS/EISソフトウェアを再起動すると、ログオンウィンドウが開きます。初めてのログオンのみ、用意されている「ADMIN(ユーザーID/パスワードとも同一)」を使用する必要がありますが、「アクセス制御設定ウィンドウ」内の「ユーザー/グループ管理情報」等の入力可能な状態となっていますので、そちらで任意のユーザーID/パスワードを設定してください(次の(3)で解説します)。

(3)ユーザーやグループの情報を設定するには(2)の作業が終了した段階で、「ユーザーの追加・編集(パスワード変更など)」作業が可能です。「アクセス制御設定ウィンドウ」の「アクセス制御の設定」にあります「ユーザー/グループ管理」を使って、アクセスコントロールを行ないたいユーザー情報を追加してください。

- ①グループの追加・編集・削除は「ユーザー/グループ管理」の「グループ管理」より行ないます。「SAS/EISソフトウェア」のメタデータやアプリケーションに対するアクセスコントロールは「グループ単位」で設定するので、各ユーザーは必ずいずれかのグループに所属させる必要があります。
- ②ユーザーの追加・編集・削除は「ユーザー/グループ管理」の「ユーザー管理」より行ないます。なお、ユーザーはいずれかのグループに所属させないと、作業が終了できないようになっています。

(4)アクセス制御リストの定義を行なうには(3)で「グループ」の設定が終了すると「SAS/EISソフトウェア」に対するアクセスコントロールが実現可能となります。「アクセス制御設定ウィンドウ」の「アクセス制御の設定」にある、「アクセス制御リストの定義」を使います。

- ①メタベースに対するアクセスコントロールの設定は「アクセス制御の設定」の「データアクセスコントロール」より行ないます。データアクセスウィンドウの「グループ」にはアクセスコントロールの対象とするグループを、「データ元」にはアクセスコントロールを行ないたいデータセットやMDDDBを指定してください。「次元:」には選択したメタベースの持つ列が、「レベル:」には列の持つ階層が表示されます。設定したい列の上で右クリックし、表示されるポップアップメニューより項目を選択してください。たとえば、「ISUSERS」に所属しているユーザーには「売り上げ予測」を非表示、「国」に係わる階層構造において「メキシコ」を初期値として表示させる設定例は以下の通りです。



- ②アプリケーションに対するアクセスコントロールの設定は「アクセス制御の設定」の「アプリケーションアクセスコントロール」より行ないます。アプリケーションアクセスウィンドウの「グループ」にはアクセスコントロールの対象とするグループを、「ライブラリ」および「アプリケーションデータベース」にはアクセスコントロールを行ないたいSAS/EISアプリケーションが登録されているカタログを指定してください。「アプリケーション:」には指定したカタログに登録されている全アプリケーションが表示されます。何らかの制限を設定したいアプリケーション上で右クリックし、表示されるポップアップメニューより項目を選択してください。たとえば、「ISUSERS」に所属しているユーザーに「GSMDDDB」「SALES」の2つのアプリケーションの実行権限を与えない設定例は以下の通りです。



③機能に対するアクセスコントロールの設定は「アクセス制御の設定」の「関数アクセスコントロール」より行ないます。機能アクセスウィンドウの「グループ」にはアクセスコントロールの対象とするグループを指定してください。


「機能:」には「SAS/EISソフトウェア」にて使用される機能一覧が表示されます。制限をかけたい機能の上で右クリックし、表示されるポップメニューより項目を選択してください。たとえば、「ISUSERS」に所属しているユーザーに、「保存」の権限を与えない設定例は以下の通りです。




(5)アクセスコントロールが有効になっているSAS/EISプログラムの実行例

たとえばデータアクセスコントロールが設定されたメタベースを使って「多次元レポート」を実行した場合、制限の有無によって以下のような違いが発生します。

アクセスコントロールが設定されていないユーザー



アクセスコントロールが設定されているユーザー



たとえば「売上げ予測」がドロップされているユーザーが多次元レポートを実行した場合、各年で「売上げ予測」の表示・非表示という差が発生しています(上記)。なお、列を「ドロップ」・「非表示」のどちらでもアプリケーション上で表示されないということは共通ですが、「平均」や「合計」などの統計量算出において、その列を計算に含む / 含まないの違いが発生します。設定を行なう際は十分にご確認ください。

2. ネットワーク上での使用に対して設定するセキュリティ

SASシステムをネットワーク上で利用する方法として、SAS/CONNECTソフトウェア、SAS/SHAREソフトウェアなどの利用が挙げられます。またSAS/IntrNetソフトウェアやEnterprise Minerソフトウェアといったプロダクトでも、ネットワーク形態がとられます。ここでは、これらのネットワーク形態におけるセキュリティ問題とその対応について説明します。

2.1 ネットワーク間接続情報に関するセキュリティ

クライアント・サーバー形態をとり、ネットワーク環境で相互のマシンと連携して処理を行なう場合、必ずそのマシン間に何らかの情報のやり取りが発生します。1つの例として「SAS/CONNECTソフトウェア」を使ってローカルマシン(Windowsマシン)からリモートマシン(UNIXマシン)への接続を行なったときの通信情報を、SPAWNERプログラムを起動しない場合と起動する場合について検証してみましょう。

SPAWNERプログラムを起動していない場合

[テスト時の状況]

- ローカルマシン Windowsマシン(マシン名:jpnkcy)
- リモートマシン UNIX(Sun OS Xマシン名:SUN450)
- 通信メソッド COMAMID=TCP
- 上記以外はデフォルト設定のまま接続

接続時に用いたプログラム

```

/* ブロック① */
FILENAME rlink '!sasroot\connect\%saslink%\tcpunix.scr';
OPTIONS COMAMID=TCP REMOTE=SUN450;
SIGNON;

/* ブロック② */
RSUBMIT;
DATA TEST;
  A = 'CONNECT TEST';
RUN;
PROC PRINT; RUN;
ENDRSUBMIT;

/* ブロック③ */
SIGNOFF;
    
```

上記のプログラムを実行した場合、

- ・ブロック①/③に関しては「TELNET(ポート番号23)」を使用して通信を行なう
- ・ブロック②に関してはSASシステムが任意のポートを割り当てて通信を行なうための通信状況を確認することができます。LANアナライザを使用してその状況を取得した結果は次の通りです。

ブロック①/③についての情報



## ブロック②についての情報



上記のように、TELNETを利用して接続しているブロック①/③では、リモートマシン(UNIXマシン)に対する「ユーザーID/パスワード」が明らかに確認できます。またSAS/CONNECTソフトウェアのセッション確立中のブロック②についても、多少の工夫によりプログラムや実行結果を復元することが可能です。これらの情報に対して少しでも防御する手立てはないものでしょうか。1つの手段としてリモートマシンへの接続の際に「SPAWNERプログラム」を使用することが挙げられます。SPAWNERプログラムは通常SAS/CONNECTソフトウェアの一機能として提供されており、下記の作業によって使用環境を調整できます。

- ・リモートマシンにおけるSPAWNERプログラムの起動(UNIXマシンの場合)  
例) SASROOT/utilities/bin/sastcpd -service サービス名  
-background -shell
- ・スクリプトファイルの手直し  
UNIXマシンに対する接続の場合、ローカルマシン側の接続スクリプトファイルにおける「unxspawn:」以下と「signoff:」以下を手直しする必要があります。

ではSPAWNERプログラムが起動しているマシンにSAS/CONNECTソフトウェアを使って接続を行ない、その違いを見てみましょう。

## SPAWNERプログラムを起動している場合

テスト時の状況

ローカルマシン Windowsマシン(マシン名:jpnkcy)  
リモートマシン UNIX(Sun OS)マシン名:jpnsun)  
通信メソッド COMAMID=TCP  
接続時のポート番号を5000に固定

接続時に用いたプログラム


```
/* ブロック① */
%LET host=172.26.20.86 5000;
FILENAME rlink '!sasroot%connect%yasalink%tcpunix.scr' ;
OPTIONS COMAMID=TCP REMOTE=host ;
SIGNON ;

/* ブロック② */
RSUBMIT ;
DATA TEST ;
  A = 'CONNECT TEST' ;
RUN ;
PROC PRINT; RUN;
ENDRSUBMIT ;

/* ブロック③ */
SIGNOFF ;
```

このプログラムは全ブロックとも「SPAWNERプログラム」を介して通信が行なわれ、リモートマシン側との通信は「ポート番号5000」を使用する例です。SPAWNERプログラムをポート番号5000で起動する方法は後述します。

## ブロック①についての情報



上記のように、SPAWNERプログラムの利用でTELNET部分における情報漏洩に対して一定の手段を講じることができます。一方、この場で掲載していない「ブロック②」のSAS/CONNECTソフトウェアセッション確立中の情報は、SPAWNERプログラムでもその漏洩防止に役立てることが残念ながらできません。この部分に対して手段を講じるためには「SAS/SECUREソフトウェア」を使用して暗号化を行なうことが考えられます(現時点で日本市場への製品出荷はされていません)。

## 2.2 リモートマシン(サーバマシン)接続に対するセキュリティ対応

通信情報からの漏洩対策としてSPAWNERプログラムを使用することは前で述べた通りですが、もう一点、サーバマシンの不正利用に関する対策を講じなければなりません。不正利用対策としては「サーバマシンへのアクセスに制限を加える」といった方法が効果的です。具体的には「サーバマシンに接続してそのリソースを使用するプログラムに対する制限」すなわち「ポート番号」に対する適切な制限です。これらを実現するためには下記の対策が一般的です。

- ・サーバマシンにおける使用可能なポート番号の設定
- ・ファイアウォールを設置した、想定外のポート番号使用アプリケーションの稼働抑止

この節の「(1)ポート番号の制限への対応策」では、それらの対策に対応するためにSASシステム側が提供している機能を説明します(なお、今回の記事はCOMAMID=TCPのTCP/IP通信を使用していることを前提としています)。また、サーバマシン利用時の「サーバマシンへのログオン」制限によるセキュリティ強化についてもこの節の「(2)ユーザー認証の設定」で説明します。これについてはWindowsマシンを例に若干触れたいと思います。

## (1)ポート番号の制限への対応策

「2.1ネットワーク間接続情報に関するセキュリティ」での「SPAWNERプログラムを起動していない場合」での通信情報におけるポート番号変移を、以下の図に示します(接続開始時にTELNETを利用している場合)。

## ブロック① SAS/CONNECTソフトウェアセッション確立時



ブロック② SAS/CONNECTソフトウェアセッション確立後(処理中)

```
TELNET Telnet Data ...
TCP 2195 > 51948 [SYN] Seq=3134837262 Ack=0 Wf=8192 Len=0
TCP 51948 > 2195 [SYN, ACK] Seq=424999920 Ack=3134837262 Wf=24820 Len=0
TCP 2195 > 51948 [ACK] Seq=3134837263 Ack=424999921 Wf=8760 Len=0
TCP 51948 > 2195 [PSH, ACK] Seq=3134837263 Ack=424999921 Wf=8760 Len=0
TCP 2195 > 51948 [ACK] Seq=424999921 Ack=3134837447 Wf=31580 Len=0
TCP 51948 > 2195 [PSH, ACK] Seq=424999921 Ack=3134837447 Wf=31580 Len=0
TCP 2195 > 51948 [ACK] Seq=424975005 Ack=3134837505 Wf=8576 Len=0
TCP 51948 > 2195 [PSH, ACK] Seq=3134837503 Ack=424975006 Wf=8512 Len=0
TCP 2195 > 51948 [ACK] Seq=3134837511 Ack=424975006 Wf=8512 Len=0
TCP 51948 > 2195 [ACK] Seq=424975006 Ack=3134837607 Wf=31580 Len=0
TCP 2195 > 51948 [PSH, ACK] Seq=424975006 Ack=3134837607 Wf=31580 Len=0
```

上記の図の中の数字は、それぞれ下記のSASシステムの利用を表します。

	ポート番号	ポート機能を使用しているSASシステム
ブロック①	2554	ローカルマシン
	telnet(23)	リモートマシン
ブロック②	2595	ローカルマシン
	53948	リモートマシン

ブロック①ではTELNETポートを使って通信を行なうのは既に説明した通りです。ブロック②ではローカル・リモートの双方でウェルノーンポート以降(1024番以降)の任意の空きポート番号を使用します。この状況では、リモートマシンもしくはファイアウォールマシンにおける使用ポート制限が適切に設定できません。この使用ポート番号を固定化するために必要なポイントは以下の通りになります。

- ・SAS/CONNECTソフトウェア接続開始時・接続終了時の使用ポート番号の指定
- ・SAS/CONNECTソフトウェアセッション確立中のポート番号の指定

SAS/CONNECTソフトウェア接続開始時・接続終了時の使用ポート番号の指定

接続時のプログラムを以下のように記述します。

```
%LET host=172.26.20.86 5000;
FILENAME rlink '!sasroot\connect\%saslink\%tcpunix.
scr' ;
OPTIONS COMAMID=TCP REMOTE=host ;
SIGNON ;
```

このケースはリモートマシン側のTELNETポートを一般的なポート番号23以外に変更する場合です。上記のように「%LET」を用いて、「IPアドレスポート番号」を指定しておくことで、接続開始時・接続時に使用するポート番号を変更できます。なお、当然リモートマシン側では、TELNETサービスを指定されたポート番号(上記の場合は5000)で起動しておく必要があります。SPAWNERプログラムを使用した場合も、上記と同様の記述が必要な場合があります。SPAWNERプログラムを「-SERVICE ポート番号」オプションを使って起動した場合、「-SERVICE」オプションで指定されたポート番号を使って接続開始・終了およびSAS/CONNECTソフトウェアの通信が行なわれるためです。今回はSPAWNERプログラム起動時に「-SERVICE 5000(あるいはサービス名)」が指定されている場合です。

例)

```
!SASROOT/utilities/bin/sastcpd -service spawner
-background -shell
```

上記はUNIXマシンにおけるSPAWNERプログラム起動方法です。サービスとして「spawner」には5000番ポートがservicesファイルで割り当てられていると仮定します。

では、このSPAWNERプログラムを上記の例の設定を使って起動し、SAS/CONNECTソフトウェアを使用した場合、ポート情報は通信情報上どのようにになっているでしょう。

```
TCP 1098 > 5000 [SYN] Seq=391881199 Ack=0 Wf=8192 Len=0
TCP 5000 > 1098 [SYN, ACK] Seq=4139115911 Ack=391881200 Wf=8760 Len=0
TCP 1098 > 5000 [ACK] Seq=391881200 Ack=4139115911 Wf=8760 Len=0
TCP 5000 > 1098 [PSH, ACK] Seq=4139115914 Ack=391881200 Wf=8760 Len=3
STH Conf. Asset = 32768/20:10:07:10:33:10 Cost = 10 Port = 2x9000
TCP 1098 > 5000 [PSH, ACK] Seq=391881200 Ack=4139115917 Wf=8717 Len=3
TCP 5000 > 1098 [ACK] Seq=4139121917 Ack=391881200 Wf=8760 Len=0
TCP 5000 > 1098 [PSH, ACK] Seq=4139121917 Ack=391881200 Wf=8760 Len=3
TCP 1098 > 5000 [PSH, ACK] Seq=391881200 Ack=4139115960 Wf=8714 Len=3
TCP 5000 > 1098 [PSH, ACK] Seq=4139115960 Ack=391881200 Wf=8760 Len=3
```

ポート番号1098はローカルマシンのSASシステム、ポート番号5000はリモートマシンのSASシステムを表します。接続開始時およびSAS/CONNECTソフトウェアセッション確立時を問わず、ポート番号の利用状況は変わりません。リモートマシン側のポート番号が固定されるので、ポート番号の制限が容易になります。

SAS/CONNECTソフトウェアセッション確立中のポート番号の指定

リモートマシンのSASシステムの起動時オプションに「PORTFIRST」「PORTLAST」システムオプションを使用します。

```
'sas -set PORTFIRST 5000 -set PORTLAST 5100 ..... ' LF;
```

SAS/CONNECTソフトウェアの接続開始時・終了時に「TELNET」を使用している場合、上記のオプション利用は非常に意味があります。一方、SPAWNERプログラムは利用するポート番号が一定になるので無意味です。

SPAWNERプログラム起動時に「INHERITANCEオプション」が設定されている場合(バージョン8でのデフォルト)。バージョン6では「NOINHERITANCEオプション」がデフォルトなので、上記のオプションを使用し、ポート番号を指定することが有効なケースがあります。

SAS/CONNECTソフトウェアで使用するスクリプトファイルにて、「PORTFIRSTポート番号」「PORTLASTポート番号」を追加することで、SAS/CONNECTソフトウェアの通信をこの範囲内に納めることが可能となります。上記の例では「5000～5100」の任意の番号で通信が行なわれることを意味しており、もし「5000」のみを使用したい場合は、「PORTFIRST」「PORTLAST」にて同一の番号を指定します。ただし、SAS/CONNECTソフトウェアは1セッションで1ポートを占有するため、「5000」のみに限定した場合、同時接続できるユーザーは1ユーザーのみとなります。必ず想定される「同時接続ユーザー数」分のポート数を範囲として指定してください。

(2)ユーザー認証の設定

サーバーマシンがWindows(NT, 2000)の場合、SAS/CONNECTソフトウェアを利用する際に「ドメイン認証を行なう」設定でSPAWNERプログラムを起動すれば、サーバーマシンへの接続時にユーザーID/パスワードによる認証が可能です。

例)

事前に「SPAWNER」サービスの停止およびサービスからの登録削除が必要です。

- a) 接続するユーザーに対して、サーバー側にて「バッチジョブとしてログオン」の権限を設定します。

【Windows NTの場合】

- スタートメニュー 「管理ツール」 「ユーザーマネージャ」を選択してください
- 「原則(P)」 「ユーザーの権利(U)」を選択後、「高度なユーザー権利の表示(S)」をチェックしてください
- 「権利(T)」の項目で、「バッチジョブとしてログオン」を選択し、「追加(A)」をクリックして接続するユーザー(グループ)を選択してください

iv) この権利を与えられたアカウント(G)に追加したユーザーが登録されていれば、正しく設定されています

【Windows2000の場合】

- i) スタートメニュー「設定」「コントロールパネル」「管理ツール」「ローカルセキュリティポリシー」を選択してください
- ii) 「セキュリティの設定」「ローカルポリシー」「ユーザーの権利の割り当て」を選択後、「バッチジョブとしてログオン」をクリックしてください
- iii) 「追加(A)」をクリックして接続するユーザー(グループ)を選択してください
- iv) 「バッチジョブとしてログオン」のポリシーに追加したユーザーが登録されていれば、正しく設定されています

b) SPAWNERプログラムをサービスに登録します。「コマンドプロンプト」を開き、以下のコマンドを入力してください。

```
!SASROOT%spawner -i -c tcp -security -authserver
DOMAIN名(サーバー名)
```

V6では「-authserver」以降に「サーバー名」を指定できません。

c) WindowsNT/2000のサービスとしてSPAWNERが登録されるので「サービスの開始」を行なってください。

設定終了後、SAS/CONNECTソフトウェアの接続ではドメインを使ったユーザー認証が行なわれるようになります。つまり「ドメインサーバー」に対しての問い合わせが「SAS/CONNECTソフトウェアの接続開始時」および「接続終了時」に発生するようになりますので、注意が必要です(たとえば「ドメイン名:DOMAIN.COM / ユーザーID:USER1」のユーザーがSAS/CONNECTソフトウェアを使って「サーバーマシン名:SERVER\_SAS (spawner -i -c tcp -security -authserver DOMAIN.COMでSPAWNER起動)」へ接続しても、ドメイン(DOMAIN.COM)への参加の有無にかかわらず、SAS/CONNECTソフトウェアによる接続を行なった時点で、ドメインサーバーへの認証作業が発生します)。

### 3. SAS/IntrNetソフトウェアにおけるセキュリティ

SAS/IntrNetソフトウェアに関しては、今まで説明したデータセット、SAS/EISソフトウェアに対するセキュリティ、SAS/CONNECTソフトウェアに対するセキュリティと若干要求されるものが異なり、不特定多数のユーザーに対してのセキュリティを考慮する必要があります。最後にこのソフトウェアのセキュリティに関する設定事項について簡単にご説明します。

#### 3.1 AUTHLIB機能を使ったアクセス制限

リリース8.2のSAS/IntrNetソフトウェアには、ライブラリ、メンバ、エンリタイプごとにアクセスの制限を設定する「AUTHLIB」機能が追加されました。これによりユーザーごとにアクセス可能なリソースを制御できます。ただし、「AUTHLIB」機能はSAS/EISソフトウェアの「アクセスコントロール」と異なり、あくまでも「プログラム」にて対応する必要があります。

##### (1) 事前に準備しておくべきこと

AUTHLIB機能を使用するにあたって、アクセス権限を設定するための「AUTHLIBデータセット」を用意しておく必要があります。サンプルのデータセットは「SASHELP.AUTHLIB」として提供されています。データセットのレイアウトは以下の通りです。

列名	属性	長さ	解説
Rule	文字	7Byte	「INCLUDE」(許可)か「EXCLUDE」(禁止)
Libname	文字	8Byte	SASデータライブラリ名
Memname	文字	32Byte	SASデータセット名やSASCATALOG名など
Memtype	文字	8Byte	「DATA」「CATALOG」「VIEW」など
Objname	文字	32Byte	主にSASCATALOG内のエンリ名
Objtype	文字	8Byte	エンリタイプ(SCL、PROGRAMなど)
Comment	文字	128Byte	注記すべき項目など任意

たとえば「SASHELP.AUTHLIB」データセットを以下の通り設定したと仮定します。

Rule	Libname	Memname	Memtype	Objname	Objtype
INCLUDE	SASHELP	*	DATA	*	*
INCLUDE	SASHELP	*	VIEW	*	*
INCLUDE	SASHELP	*	MDDDB	*	*
INCLUDE	SASDAT	*	*	*	*
INCLUDE	SASDAT	MYCAT	CATALOG	*	*

この場合、INCLUDE指定のある下記のリソースについてはアクセス権限が与えられます。

- ・SASHELPライブラリ内の「データセット」「ビュー」「MDDDB」
- ・SAMPDATライブラリ内の「MYCATカタログ」を除く全リソース

EXCLUDE指定のあるリソース、およびINCLUDE指定以外のライブラリ内リソース(下記に示します)はアクセス権限が与えられません。

- ・SASHELPライブラリ内の「データセット」「ビュー」「MDDDB」以外の全リソース
- ・SAMPDATライブラリ内の「MYCATカタログ」



## (2) AUTHLIB機能を使ったプログラムの作成例

「AUTHLIB」機能はあくまでも「権限の有無についての情報を返す」だけであり、実際に権限のないライブラリを参照することは可能です。よって以下のマクロプログラムを作成し、アクセス権限の有無によって処理を分岐させることで利用できます。

```
DATA _null_ ;
/* アクセス権限の判定 */
IF (APPSRV_AUTHLIB('sashelp','*', 'data'))
THEN DO ;
/* 権限のある場合 */
CALL SYMPUT('authok','1');
END ;
ELSE DO;
/* 権限のない場合 */
CALL SYMPUT('authok','0');
END ;
RUN ;

%MACRO authchk;
%IF "&authok" = "1" %THEN %DO;
/* アクセス権限がある場合の処理 */
proc print data=sashelp.class;
run;
%END ;
%ELSE %DO;
/* アクセス権限がない場合の処理 */
%PUT "参照権限がありません";
%END;
%MEND authchk;

%authchk;
```

上記の「APPSRV\_AUTHLIB」関数は、下記のステートメントにて指定されたリソースのアクセス権限取得が可能です。rcの値が「0」の場合はアクセス権限なし、「1」の場合はありとなります。

```
rc = APPSRV_AUTHLIB('ライブラリ名', 'メンバ名', 'メンバタイプ', 'オブジェクト名', 'オブジェクトタイプ');
```

## (3) ユーザーごとにアクセス権限を変更する場合

AUTHLIB機能を使う場合、各リソースに対するアクセス権限は「SASHELP.AUTHLIBデータセット」より情報を取得するのがデフォルト設定ですが、「APPSRV\_AUTHDS」関数によってそのデータセットを変更することが可能です。


```
rc = APPSRV_AUTHDS('データセット名');
```

各ユーザーが利用を開始するに当たり上記のプログラムを実行し、各ユーザー用に設定したAUTHLIBデータセットを有効にすれば、ユーザーごとにアクセス権を変更できます。

## 3.2 使用するポート番号の制御について

SAS/IntrNetソフトウェアにおいてもTCP/IP通信方式を使うソフトウェアである以上、ポート番号の使用に対するセキュリティは必ず考慮すべきものとなります。SAS/IntrNetソフトウェア リリース8.2において、指定できるポートと各ソフトウェアの構成は以下の通りとなります。

イメージ図



上記は「ロードマネージャー」+「ソケットサービス」を用いたSAS/IntrNetソフトウェア環境の場合で、「ロードマネージャー」と「ソケットサービス」がそれぞれ異なるマシンで稼働している例です。なお、プールサービスに関しては、下記の点を考慮すれば、ポート番号の設定が可能です。今回は紙面の都合上紹介いたしません、ソケットサービスの設定をご参考ください。

- ・アプリケーションサーバー(プールサービス稼働)とロードマネージャーの起動マシンが違う場合、アプリケーションサーバー側に「Spawnerプログラム」を起動が必要な点
- ・プールサービス起動時のポート番号は動的に割り当てることが必要であり(SAS/IntrNetソフトウェアの不具合)、ポート番号を静的に指定して運用することが困難


これらのポート番号を指定するのは、WWWサーバー内の実行スクリプト(CGIプログラム)が格納されているフォルダ(ディレクトリ:「scriptsやcgi-bin」などの名称)に存在する「broker.cfg」です。

Broker.cfgの設定例


```
SocketService default "Reuse existing session"
ServiceDescription "Pages reference this generic
server when they +
don't care which service is used."
ServiceAdmin "[your-name]"
ServiceAdminMail "[your-email]@[your-site]"
Server Apsrv
Port 5011 5012 5013 /* ソケットサービス起動ポート番号 */
ServiceTimeout 30
/* ロードマネージャー起動ポート番号 */
ServiceLoadManager WWWsrv:5555
# Remove the following line for any servers before
V8.1
FullDuplex True /* 設定は「True」 */
```

最後の「FullDuplex True」の指定は重要です。この設定がない場合、「ロードマネージャー」「ソケットサービス(アプリケーションサーバー)」の通信使用ポートにおいて、指定のポート番号で戻ってこない場合があります。






上記が「FullDuplex False」を指定している場合です。ロードマネージャーからソケットサービスへの通信はポート番号「5555」を使用しているものの、ソケットサービスからの戻りが「5555」以外の場合があります。もし、この「ロードマネージャー」「ソケットサービス」の間に「ファイアウォール」を設置している場合、「ポート番号5555」で設定を行なうことができなくなるため注意が必要です。上記の「Broker.cfg」で稼動している場合の例は、以下の通りになります。



また「ロードマネージャー」「アプリケーションサーバー」が同一マシンに存在している場合、このような状況になります。

前述の「Broker.cfg」において「ServiceLoadManager Apsrv:5555」となっている例です。



「任意のポート番号」と記載してある部分は、ポート番号が実行ごとに異なることを示します。

### おわりに

今回の特集では「リソースに関するセキュリティ機能の利用」と「ネットワーク上の通信情報に関する対策」をメインにご説明しましたが、最初にご説明したOS自体が持つセキュリティ機能をまず活用することが重要です。SASシステムの各プロダクトを使う上で、利用している他の製品が持つセキュリティ機能の設定なども併せて利用するケースもあります。利用環境に最適なセキュリティ環境を構築する上で、是非記事をご参考ください。



# Q&A

TABULATEプロシジャで、存在しない ICLASS変数の値を表に含める (V8) 文字列が8バイト以上ある変数でTRANSPOSEプロシジャを利用する 起動時の環境設定の内容を確認する

2つのSASデータセットのうち、片方に存在するデータを もう片方から削除する方法

SAS起動時にエクスペローラを表示しない設定

HTMLファイルを直接Webサーバにアップロードする方法

SAS/IntrNetソフトウェアで、ユーザの入力情報を

SASデータセットに保存する方法

HTMLファイルにグラフを段組で出力する方法

SASのバージョンを検知する方法


MP CONNECTでの多重処理時のプロセス

JMPソフトウェア Version 4とLOGISTICプロシジャの

パラメータの信頼区間について

n が大きいときの順列・組み合わせの計算方法

リソースをなるべく使用しない IIMLの計算方法



**Q** TABULATEプロシジャで多次元の表を作成するとき、存在しないクラス変数の値を表に含めることはできますか？  
下の例で、表1の「SHOP=東京2」の表に「CODE=S02」の行と「DATE=2001-01-02」の列を追加して、表2のような出力にしたいと思います。

表1

shop 東京1			
		date	
		01-01-01	01-01-02
		数量	数量
code			
S01	3	1	
S02	1	1	
S03	2	1	

shop 東京2			
		date	
		01-01-01	
		数量	
code			

		date	
		01-01-01	01-01-02
		数量	数量
code			
S01	3	.	
S02	.	.	
S03	2	.	

表2

shop 東京2			
		date	
		01-01-01	01-01-02
		数量	数量
code			
S01	3	.	
S02	.	.	
S03	2	.	

**A** SASシステムバージョン8から追加されたCLASSDATAオプションを使って、出力するクラス変数の値を制御できます。表に含めるクラス変数の値の組み合わせをデータセットにしておき、TABULATEプロシジャで「CLASSDATA=データセット名」を指定します。ご質問の例の場合、クラス変数SHOP、CODE、DATEの全組み合わせを含めた次のようなCLASSDATAデータセットを作ります。

CLASSDATAデータセット

OBS	shop	code	date
1	東京1	S01	01-01-01
2	東京1	S02	01-01-01
3	東京1	S03	01-01-01
4	東京2	S01	01-01-01
5	東京2	S02	01-01-01
6	東京2	S03	01-01-01
7	東京1	S01	01-01-02
8	東京1	S02	01-01-02
9	東京1	S03	01-01-02
10	東京2	S01	01-01-02
11	東京2	S02	01-01-02
12	東京2	S03	01-01-02

プログラム例

```
options pageno=1;
/* サンプルデータセットの作成 */
data sales;
```

```

input shop $ code $ date yymmdd10. +1 suryo;
format date yymmdd8.;
cards;
東京1 S01 2001/01/01 3
東京1 S02 2001/01/01 1
東京1 S03 2001/01/01 2
東京2 S01 2001/01/01 3
東京2 S03 2001/01/01 2
東京1 S01 2001/01/02 1
東京1 S02 2001/01/02 1
東京1 S03 2001/01/02 1
;
run;

/* CLASSDATAデータセットの作成 */

proc sql;
create table _class as
select * from
(select distinct(shop) from sales) ,
(select distinct(code) from sales) ,
(select distinct(date) from sales) ;
quit;

proc tabulate data=sales classdata=_class ;
class shop code date;
var suryo;
tables shop,code,date*suryo='数量'*f=10. ;
keylabel sum=' ';
run;

```

**Q** 1レコードが複数行にまたがるデータを入力し、TRANSPOSEプロシジャを利用して1行ずつのレコードにまとめる事を考えています。しかし、転置のキーとなる変数が8バイトより長く、8バイト以下の長さでは値が重複してしまい、うまく転置がされません。また、該当する変数をLABELとして設定したいのですが、8文字で切られているために思うような結果になりません。何か回避方法はあるでしょうか。

データの例

```

aaaaaaaa 02/02/27 koumoku9 09123456789
aaaaaaaa 02/02/27 koumoku10 10123456789
aaaaaaaa 02/02/27 koumoku11 11123456789
bbbbbbbb 02/02/27 koumoku9 09234567890

```

TRANSPOSE後のイメージ

INPUT_DATA_GRP	DATE	koumoku9	koumoku10	koumoku11
aaaaaaaa	02/02/27	09123456789	10123456789	11123456789
bbbbbbbb	02/02/27	09234567890		

**A** SASシステム バージョン6では、変数名の長さは8バイトまでの制約があります。そのため、ご指摘のようにバージョン6のTRANSPOSEプロシジャにおいても、IDおよびIDLABELに指定可能な変数は8文字までとなっています。ただし、設定したい変数をいったんユニークな8文字以内の別の変数に保管し、その変数をIDとして使用することにより同等の結果を実現できます。なお、新バージョンとなるSASシステム バージョン8eからは、変数名が32バイトまで、変数ラベルは256バイトまでに拡張されており、制約が大幅に緩和されています。下記例をご覧ください。

```

/* TESTデータ作成 */
data testtr;
label a = 'input_data_grp'
      b = 'input_data_date'
      c = 'input_data_koumoku'
      d = 'input_data_count' ;
format a $9. b yymmdd8. c $9. d $12. ;
input @1 a $9. @11 b yymmdd8. @20 c $9. @30 d 12. ;

datalines;
aaaaaaaa 02/02/27 koumoku9 09123456789
aaaaaaaa 02/02/27 koumoku10 10123456789
aaaaaaaa 02/02/27 koumoku11 11123456789
bbbbbbbb 02/02/27 koumoku9 09234567890
;
run;

/* ユニークな8バイト以内の別変数を作成 */
data testtr1;
set testtr;
e = compress(substr(c,1,3)||substr(c,6,4)) ;
run;

/* IDへユニークな別変数を指定 */
proc transpose data=testtr1 out=trans1(drop=_name_)
;
by a b ;
id e ;
idlabel c ;
var d ;
run;

```

**Q** 環境設定ファイル(SASシステム バージョン8の場合はsasv8.cfg)をカスタマイズしたのですが、この設定が反映されているか分かりません。SASシステム起動時に、設定についての情報を一括して把握する方法はありますか。

**A** SASシステムを起動する際のコマンドに、SASシステム オプションVERBOSEを指定すれば可能です。この情報は、ログウィンドウまたは標準出力に出力されます。

起動コマンドの例

UNIXの場合	\$ sas -verbose
Windowsの場合	> c:\yprogra~1\sasins~1\sas\y8\ylnls\yja\y sas.exe -verbose

出力の例[ Solaris版SASシステム リリース8.2日本語版の場合 ]

```

Option          Value
=====
CONFIG          /home/sas82/sasv8.cfg
DBCSLANG        JAPANESE
DBCSTYPE        EUC
EMAILSYS        !SASROOT/utilities/bin/sasmailer
FILELOCKS       FAIL
FSIMM           "x11=pipe"
FSIMMOPT        pipe:motif
MAXMEMQUERY     6291456
PATH            !SASROOT/../../../../dbcs/sasexe
                !SASROOT/../../../../sasexe
SET             [SASROOT = /usr/local/sas82/nls/ja]
                [GFONT0 = !SASROOT/kanji]
                [GFONT1 = !SASROOT/mincho]
                [sasroot = /usr/local/sas82/nls/ja]
WORK            /tmp
DBCS            ON
ENCODING        EUC-JP
LOCALE          JAPANESE
MEMSIZE         64M
MSG             '!SASROOT/sasmsg'
                '!SASROOT/../../../../sasmsg'
MSGCASE         OFF
SETJMP          ON
STIMER          ON
VERBOSE         ON
XCMD            ON

WORK            /tmp
MAPS            !SASROOT/../../../../maps
SASHELP         '!SASROOT/sascfg'
                '!SASROOT/../../../../sascfg'
                '!SASROOT/../../../../sashelp'
SASAUTOS        !SASROOT/../../../../sasautos
HELPLLOC        '!sasuser/classdoc'
                '!SASROOT/../../../../X11/native_help'
NODMS
SASUSER         /home/sas82/sasuser
NEWS            !SASROOT/../../../../misc/base/news
APPLETLOC       !SASROOT/../../../../misc/applets
DOCLOC          !SASROOT/../../../../install/docloc.htm
NOASYNCHIO
SORTSIZE        48M
MSYMTABMAX      4M
MVARSIZE        32K
DMSEXP
SASSCRIPT       '!SASROOT/../../../../misc/connect'
UNIVERSALPRINT
SYNCHIO
NODMSBATCH
NOBOOTSTRAP
TERMINAL

```

**Q** 2つのSASデータセットがあります。片方のデータセットに含まれるレコード( オブザベーション )を、もう1つのSASデータセットから削除したいと思います。何かよい方法がありますか？

**A** SQLプロシジャのをEXCEPT演算子を使用すると可能です。下記の例は、TRANSデータセットに存在するオブザベーションをMASTERデータセットより削除するプログラムです。

プログラム例

```

data master;
    input ID NAME $;
cards;
1 A
2 B
3 C
4 D
5 E
6 F
7 G
;
run;

data trans;
    input ID NAME $;
cards;
2 B
3 C
5 E
;
run;

proc sql;
    create table master as
    select * from master
    except
    select * from trans
;
quit;

```

**Q** SAS起動時にエクスプローラが自動的に表示されますが、普段は使用しません。これを表示しないように設定することはできますか？

**A** Windows版同様、UNIX版SASシステム リリース8.2においても、初期設定では、SASシステム起動と同時にエクスプローラが起動されるように設定されています。SASシステム起動時にエクスプローラを起動させないためには、現在の環境設定ファイル( 通常、!sasroot/nls/ja/sasv8.cfg 等 )を以下のように変更します。

変更前	変更前
-dmsexp	-nodmsexp
	-dms

**Q** ODS ( Output Delivery System )によるHTMLの出力を、直接Webサーバにアップロードできますか？

**A** FILENAMEステートメントでFTPデバイスタイプを指定すると、FTP経由でWebサーバにアップロードできます。下記に例を示します。

#### プログラム例

```
filename temp ftp '~sasdemo/public_html/class.html'
  user='user_id' host='www.sas.com'
  pass='password'
  recfm=s
  prompt debug;

ods html body=temp;

proc print data=sashelp.class;
run;

ods html close;
```

**Q** SAS/IntrNetソフトウェア リリース8.2を使用して、アプリケーションを作成しています。フォームを利用してユーザの入力情報を収集し、アプリケーションサーバに渡しているのですが、その情報をSASデータセットに保存することはできますか？

**A** SAS/IntrNetソフトウェア リリース8.2の新機能である STATISTICSオプションを利用すれば、ユーザの入力情報をSASデータセットに保存できます。具体的な手順は下記の通りです。

#### 1. STATISTICSデータセットの作成

SASシステムにて、下記のプログラムを実行します。

```
/* STATISTICSデータセットの保存先を指定 */
libname statlib "path";

/* STATISTICSデータセットを作成 */
proc appsrv port=0;
  statistics create=statlib.stat;
run;

/* STATISTICSデータセットの修正 */
data statlib.template statlib.stat;
  set statlib.stat;
  attrib dest length=$8 label='destination';
  stop;
run;
```

アプリケーションにて使用している"DEST"の値を保存することを前提にしています。

#### 2. STATISTICSステートメントを追加

アプリケーションサーバ起動時に使用する「appsrv.sas」(設定によって名前は異なる)内のAPPSRVプロシジャに「STATISTICSステートメント」を追加します。

```
proc appsrv port=0 unsafe='&';%''
  .... 省略 ....
  allocate library statlib "PATH";
  datalibs statlib;
  statistics data=statlib.stat addport
  template=statlib.template;
  .... 省略 ....
```

ソケットサービスが1つの場合は、ADDPORTオプションは不要ですが、ソケットサービスが複数起動あるいはプールサービスを使用して複数のセッションを起動する場合は、ADDPORTを使用してください。

**Q** SASシステム リリース8.2でODSを使用し、HTMLファイルにグラフを貼り付けて公開しています。都合上以下のような「段組」の表を作成し、それぞれにグラフを貼り付けようと思っておりますが、良い方法はありませんか。

#### 段組のイメージ

graph1	graph2
graph3	graph4

**A** SAS/GRAPHソフトウェアの「DEVICE=WEBFRAME」のようなグラフィックデバイスを使って簡単に上記の表を作成する方法、ODSステートメントを使って簡単に作成する方法は残念ながらありません。下のようなプログラムを使っての出力を検討してください。

#### マクロプログラム

```
/* グラフを出力するに当たって、初期化を行うマクロ(gtableinit) */
%MACRO gtableinit(cat = work.gseg,
  across = 2,
  pagewidth = 640,
  pad = 10,
  dirpath = D:\web%,
  body = gtable.html,
  title = 段組グラフ出力の例,
  webfont = MS ゴシック,
  tableborder = 2);

  %GLOBAL xpixels ypixels gifdir gcat cells cellpadding
  tablettitle htmlbody
  face border;
  %LET gifdir = &dirpath;
  %LET gcat = &cat;
  %LET cells = &across;
  %LET cellpadding = &pad;
  %LET tablettitle = &title;
```

```

%LET face      = &webfont;
%LET border    = &tableborder;
%LET xpixels   =
    %eval(%eval(&pagewidth-
(%eval(&pad*(%eval(&across-1)))))/&across);
%LET ypixels   = %eval(%eval(&xpixels/4)*3);

FILENAME _gtable "&dirpath&body";
FILENAME _gsfname "&dirpath";

GOPTIONS RESET      = all
           DEVICE    = gif
           XPIXELS   = &xpixels
           YPIXELS   = &ypixels
           GSFMODE   = replace
           GSFNAME   = _gsfname;

PROC GREPLAY IGOUT=&cat NOFS;
  DELETE _all_;
RUN;
QUIT;
%MEND;

/* TABLEタグを使って、HTMLに表を作成するマクロ(gtable) */
%MACRO gtable;
  PROC CATALOG CAT=&gcat ENTRYTYPE=grseg;
    CONTENTS OUT=_templist;
  RUN;
  PROC SORT DATA=_templist;
    BY moddate;
  RUN;
  DATA _null_;
    LENGTH name $256;
    FILE _gtable;
    SET _templist END=eof;
    IF _N_=1 THEN DO;
      PUT "<HTML>/"
        "<!-- Generated by GTABLE Macro -->/"
        "<HEAD>/"
        "<TITLE>&tabletitle</TITLE>/"
        '<META http-equiv="Content-type" '
        content="text/html">'/
        "</HEAD>/"
        "<BODY>/"
        "<TABLE cellspacing=&cellpad '
        border=&border>/"
        /* title */
        '<TR> <TD COLSPAN=" '
        "&cells"
        ' " ALIGN=center ><FONT face=" '
        "&face"
        ' " size="5" color="#002288">'/
        "<B>&tabletitle</B></FONT></TD></TR>";
    END;
  i+1;
  IF i=1 then PUT"<TR>";

```

```

PUT "<TD height=&ypixels width=&xpixels nowrap '
ALIGN=center>";
name=TRIM(name)||".gif";
PUT '<IMG SRC=" ' name ' "> </TD>';
IF i=&cells or eof THEN DO;
  i=0;
  PUT "</TR>";
END;
IF eof then put "</TABLE> </BODY> </HTML>";
RUN;
%MEND;

```

#### 実際の実行プログラム例

```

/* グラフ初期化 */
%gtableinit(dirpath=D:\web\,across=2,
pagewidth=640,body=2columns.htm);

/* 出力用グラフ作成 */
DATA test;
  DO hbar=1 TO 3;
    DO subgroup='A','B','C';
      sumvar=uniform(12345);
      OUTPUT;
    END;
  END;
RUN;
PROC GCHART;
  HBAR hbar / subgroup=subgroup sumvar=sumvar
nostats;

  VBAR hbar / subgroup=subgroup sumvar=sumvar;
RUN;
PROC GCHART;
  HBAR hbar / subgroup=subgroup sumvar=sumvar
nostats;

  VBAR hbar / subgroup=subgroup sumvar=sumvar;
RUN;
QUIT;

/* HTMLファイル出力 */
%gtable;

```

**Q** SASシステム リリース6.12とリリース8.2を併行して使用しています。双方でプログラムを共通化していますが、一部処理だけリリース番号によって使い分けたいと考えています。リリース番号を取得する方法はないでしょうか。

**A** 自動マクロ変数「&SYSVER」を使用することで、バージョンごとに異なる処理を実現できます。「&SYSVER」にはバージョン情報が格納されており、リリース6.12およびリリース8.2では次の値がセットされています。下記のプログラムを実行し、LOGウィンドウに表示されるリリース番号をご確認ください。

## &amp;SYSVERの使用例

```
DATA _null_ ;
    %PUT &SYSVER ;
RUN ;
```

## 実行結果

```
1 DATA _null_ ;
2     %PUT &SYSVER ;
X.XX                                この行にリリース番号が出力されます
3 RUN ;
```

なお、同じリリースで「TSレベル」を考慮する場合は、自動マクロ変数「&SYSVLONG」を利用してください。

## &amp;SYSVLONGの使用例

```
DATA _null_ ;
    %PUT &SYSVLONG ;
RUN ;
```

## 実行結果

```
7 DATA _null_ ;
8     %PUT &SYSVLONG ;
X.XX. nnnnPmmddy   この行にリリース番号・TSレベルが出力されます
9 RUN ;
```

sysvlongで取得される情報は「リリース番号情報 + TSレベル情報 + P + そのモジュールの作成年月日(mmddy)」で構成されます。「6.12.0060P042699」は「R6.12 TS060、モジュール構成日は1999/4/26」であることを表します。

MP CONNECTを利用して、複数のSASプロセスを同時に起動して使用していますが、処理後もプロセスが残ってしまいます。プロセスを消すことはできないのでしょうか？

Autosignonオプションを使用してリモートとの接続を行ない、かつ「wait = no」を設定しているときには、処理後もそのステータスをLISTTASKステートメントで確認できるようにSASのプロセスは残ります。ENDSUBMITステートメントによるリモートの処理後に、すぐにSASのプロセスを終わらせたい場合は、RSUBMITステートメントのオプションとして「persist = no」を設定してください。このようにすると、signoffが自動的に行なわれます。また、persistを指定しなくても、signoffを行えばSASのプロセスは消去されます。

**Q** SASシステムのLOGISTICプロシジャで計算したパラメータ推定値に対する信頼区間の値と、JMPソフトウェアで計算したロジスティック回帰の信頼区間の値が、同じデータを利用して分析しているのに異なります。なぜでしょうか。

**A** JMPソフトウェアでは、[分析]メニューから[モデルの当てはまり]を選択してロジスティック回帰を実行し、オプションとして[信頼区間]を選択すると「プロファイル尤度に基づく信頼区間」を出力します。一方、SAS/STATソフトウェアのLOGISTICプロシジャでは、バージョン8の場合、デフォルトでWald流の信頼区間を出力します(リリース6.12では、MODELステートメントにCLオプションを指定します)。このように、信頼区間の計算方法が異なるため、信頼区間の値が異なってきます。LOGISTICプロシジャでは、以下のプログラム例のように、MODELステートメントで「CLPARM=BOTH」を指定すると、Wald流の信頼区間とプロファイル尤度に基づく信頼区間の両方を出力できます。

```
/**2つの信頼区間を求めるプログラム例**/
proc logistic data=datal descending ;
    model resp = x1- x5 / clparm=both;
/**CLPARM=BOTHオプションの指定**/
run;
```

CLPARM=WALD でWald流のみ、CLPARM=PLでプロファイル尤度に基づく信頼区間のみを出力できます。

2つの信頼区間の計算方法については、下記のドキュメントを参照してください。

## リリース6.12

「SAS/STAT Software: Changes and Enhancements through Release 6.12」(注文番号:55779)p.456  
または

「SAS/STAT Software: Changes and Enhancements for Release 6.12」(注文番号:55657)p.416

## バージョン8以降

「SAS System Version 8 OnlineDoc」

SAS/STAT The LOGISTIC Procedure Details  
Confidence Intervals for Parameters

「SAS Technical News / 2002 Winter」には、SASシステム リリース6.12では順列や組み合わせを計算したい場合、GAMMA関数を利用するとの記事がありましたが、その通りにプログラムを書いてみたところ、順列や組み合わせの値が欠損値となってしまいました。これはなぜでしょうか？

```
data test;
n=200;
r=3;

perm=gamma(n+1)/gamma(n-r+1); /*PERM(200,3)を計算*/
comb=gamma(n+1)/gamma(r+1)/gamma(n-r+1);
/*COMB(200,3)を計算*/
run;
```

**A** GAMMA関数の引数として大きな数字(Windowsでは、170前後より大きな値)を与えると、GAMMA関数の真の値が極めて大きくなり計算することができません。このような場合には、GAMMA関数の自然対数を計算するLGAMMA関数をご利用ください。

```
data test;
n=200;
r=3;

/* Permutationの計算例 */
perm=exp(lgamma(n+1)-lgamma(n-r+1));
/* perm=7880400 */

/* Combinationの計算例 */
comb=exp(lgamma(n+1)-lgamma(r+1)-lgamma(n-r+1));
/* comb=1313400 */

/* 結果の比較。以下はv8以降で有効 */
perm2=perm(n,r);
comb2=comb(n,r);
run;
```

**Q** SAS/IMLソフトウェアを利用して、データセットから数値を読み込んで行列演算を行っていますが、たとえば7000×7000の行列X,Yを読み込んで「Z=X\*Y」を計算しようすると、次のエラーメッセージがログウィンドウに出力されてしまい、計算できません。どのような対処方法がありますか？

```
ERROR: (execution) Unable to allocate sufficient
memory.
At least 392000032 more bytes required.
```

**A** SAS/IMLソフトウェアでは、行列の成分をすべてメモリ上で保持しています。7000×7000の行列は、概算で400Mバイトほどのメモリを占有するので、ご質問のような計算をすると1Gバイト以上のメモリを使用することになり、エラーが発生します。対処法としては、Windowsであれば仮想メモリのサイズを上げる、UNIXではMEMSIZEの値を大きなものにするといった方法が考えられます。ただし、いずれにしても、大きな行列演算を直接行なうためにはメモリを大量に必要とするので、計算が正しく行なわれない場合があります。ここでは、「Z=X\*Y」を計算する際に行列Xを「区分け」してから演算を行なう方法をご紹介します。具体的には、次のような行列演算の手法を用いています。

Aをk×l-行列、Bをl×m-行列とする。  
 行列Aをk<sub>1</sub>×l、k<sub>2</sub>×l、...、k<sub>n</sub>×l-行列A<sub>1</sub>,A<sub>2</sub>,...,A<sub>n</sub>  
 (k<sub>1</sub>+k<sub>2</sub>+...+k<sub>n</sub>=k)で次のように区別する。

$$A = \begin{pmatrix} A_1 \\ A_2 \\ \vdots \\ A_n \end{pmatrix}$$

このとき、行列ABは次のように計算できる。

$$AB = \begin{pmatrix} A_1B \\ A_2B \\ \vdots \\ A_nB \end{pmatrix}$$

この方法は、行列Xの一部分のみを順次IMLで読み込んで計算するため、メモリの使用量が少なくすみ、ハードディスクの容量が十分ある場合には有効な方法です。なお、行列Zはデータセットとして保存されます。

```
/* IMLの起動 */
proc iml;

/* 行列yは全て読み込む */
use y;
read all into y;
close y;

/* 行列zをデータセットに書き出すために、*/
/* 出力先のデータセットzを作成 */
create z from y;
close z;

/* pは、一度に読み込む行列xの行数。大きすぎても小さすぎても */
/* メモリを消費してしまいます。 */
p=150;

/* 行列xxは行列xを区分けしたもの */
do i=1 to ceil(ncol(y)/p);

/* 行列xの一部を読み込む */
use x;
  read point (p*(i-1)+1: min(p*i,ncol(y)) ) into x;
close x;

/*z(の一部)を計算*/
z=x*y;
free x; /* 行列xを開放 */

/* 計算したzをデータセットzにAPPEND */
edit z;
  append from z;
close z;

free z; /* zを開放 */

end;
quit;
```



# New Publications

## 新刊マニュアルのご紹介

このたび、SASシステム バージョン8の持つ多くのプロシジャの中から、使用頻度の高いプロシジャを抜粋して、プロダクトごとに以下の4つの分冊に編成した日本語版リファレンスマニュアルを出版します。販売は2002年5月上旬を予定しております。

**SASプロシジャ リファレンス バージョン8:**  
SAS/CONNECT, SAS/ACCESS, SAS/SHARE,  
SAS/MDDB Serverソフトウェア

注文番号: 10085

価 格: 3,000円

本書で取り上げているプロシジャは下記の通りです(プロダクトの関係上、一部Base SASソフトウェアのプロシジャも解説しています)。

- |                   |                       |
|-------------------|-----------------------|
| SAS/CONNECTソフトウェア | SAS/SHAREソフトウェア       |
| ・ UPLOADプロシジャ     | ・ SERVERプロシジャ         |
| ・ DOWNLOADプロシジャ   | ・ OPERATEプロシジャ        |
| SAS/ACCESSソフトウェア  | SAS/MDDB Serverソフトウェア |
| ・ ACCESSプロシジャ     | ・ MDDBプロシジャ           |
| ・ DBLOADプロシジャ     |                       |
| ・ IMPORTプロシジャ     |                       |
| ・ EXPORTプロシジャ     |                       |

**SASプロシジャ リファレンス バージョン8:**  
SAS/ETSソフトウェア

注文番号: 10086

価 格: 2,000円

本書で取り上げているプロシジャは下記の通りです(今後改訂版において、その他のプロシジャも取り上げていく予定です)。

- |                |                |
|----------------|----------------|
| ・ ARIMAプロシジャ   | ・ FORCASTプロシジャ |
| ・ X11プロシジャ     | ・ X12プロシジャ     |
| ・ AUTOREGプロシジャ |                |

**SASプロシジャ リファレンス バージョン8:**  
SAS/GRAPHソフトウェア

注文番号: 10087

価 格: 4,000円

本書で取り上げているプロシジャは下記の通りです(今後改訂版において、その他のプロシジャも取り上げていく予定です)。

- |               |             |
|---------------|-------------|
| ・ GCHARTプロシジャ | ・ G3Dプロシジャ  |
| ・ GPLOTプロシジャ  | ・ GMAPプロシジャ |

**SASプロシジャ リファレンス バージョン8:**  
SAS/STATソフトウェア

注文番号: 10088

価 格: 2,500円

本書で取り上げているプロシジャは下記の通りです(今後改訂版において、その他のプロシジャも取り上げていく予定です)。

- |                |              |
|----------------|--------------|
| ・ GENMODプロシジャ  | ・ TTESTプロシジャ |
| ・ NLMIXEDプロシジャ | ・ FREQプロシジャ  |

これらのマニュアルは、各プロシジャの構文、および簡単なサンプルプログラムのみを掲載しております。上記マニュアルの詳細については、弊社ホームページ(<http://www.sas.com/japan/manual/>)をご覧ください。また、これらのマニュアルのPDF版は「インストレーションドキュメントメディア」に含め、SASシステムリリース 8.2をご契約されている全サイトに無償でお送りします(1契約につき1枚)。ご送付は5月末を予定しています。

### マニュアル販売係

T E L 03-3533-3835

F A X 03-3533-3781

E-mail [booksale@jpn.sas.com](mailto:booksale@jpn.sas.com)

なお、マニュアル申込用紙および最新のPublication Catalog (マニュアル案内パンフレット)は弊社ホームページ (<http://www.sas.com/japan/manual/>)にて公開しておりますので、併せてご利用ください。

2002年4月より、バージョン6関連マニュアル(英語版のみ)の大幅な値下げ、および「Books by Users」の一部値上げを実施いたしました。詳細は弊社ホームページをご覧ください。

# SAS Training

## SASトレーニングのお知らせ

SASインスティテュートジャパンでは、現在SASシステム リリース8.2を使用したトレーニングを行なっています。SASシステムのデータアクセス、管理、解析、そしてプレゼンテーションの機能をフルにご活用いただくために、トレーニングを随時開催中です。なお、日程・金額等のコース情報の詳細については、弊社ホームページ <http://www.sas.com/japan/training/> )をご参照いただくか、弊社トレーニング担当までお問い合わせください。

### 個別トレーニングのご案内

現在、お客さまの人材育成に関するニーズが多様化しています。より幅広くSASシステムをご利用いただくために、弊社では現場に密着したトレーニングを提供していきたいと考えております。お客さまのスケジュールに合わせた日程での個別トレーニングのご案内です。

[ 次のような場合、より効果的です ]

- ・ 同じトレーニングを何人も受講する必要がある。  
人数によっては格安となります。
- ・ 定期トレーニングに出席するには出張経費がかかりすぎる。  
ご予算の枠内で最適コースを実施できるようお手伝いします。
- ・ 忙しくてスケジュールが合わない。  
ご希望の内容、場所、期間、時間帯で実施できるよう調整します。  
お客さま先へのご訪問、または弊社トレーニングルームにお越しいただき実施します。

[ 基本価格 (10名様まで) ]

(1日:180,000円 ~ ) + (コースチャージ:1コースにつき30,000円 ~ )  
追加1名につき 18,000円 ~ / 1日 (消費税別途)

講師がお客さまの所へ訪問してトレーニングを行なう場合は、上記に諸経費(移動日・設定作業・宿泊費・交通費)を加算させていただきます。  
20名様以上の場合は、別途ご相談させていただきます。

[ お申し込み方法 ]

まず、お電話またはE-Mailにてお問い合わせください。日程、内容等を調整後、個別トレーニング申込用紙・参加者リスト用紙をお送りします。ご記入後、ファクシミリにてご返送ください。お見積書・トレーニング案内をご担当者さまにお送りします。

[ お支払い方法 ]

個別トレーニング終了後、ご請求書をお送りします。1ヶ月以内に弊社指定の銀行口座にお振り込みください。

### 5月より、お得なセットプラン開始

複数のコースをセットでお申し込みいただくと、格安となるセットプランができました。SASを基礎から学びたい方、GUIを順序だてて学びたい方には最適なセットプランとなっています。

「Basic Plan」

- 3コースセット: ¥100,000(税別)
- ・ 「SASプログラミング: Basic I」コース
  - ・ 「SASプログラミング: Basic II」コース
  - ・ 以下の5コースよりいずれか1コース選択  
「マクロ言語入門」コース  
「マクロ言語上級」コース  
「ODS入門」コース  
「実践データハンドリング」コース  
「Version 8 拡張点入門」コース

「Statistic GUI Plan」

- 3コースセット: ¥90,000(税別)
- ・ 「統計概論」コース
  - ・ 「SASによるデータ解析入門」コース
  - ・ 以下の3コースよりいずれか1コース選択  
「時系列データ解析入門」コース  
「アンケートデータ解析入門」コース  
「ビジュアルデータ解析」コース

請求書は、1コース受講終了後に発行いたします。  
日程・コースの変更は可能ですが、キャンセルはできませんのでご了承ください。

[ セットプランのお申し込み方法 ]

弊社指定のトレーニング受講申込書にありますがセットプラン名の欄をチェックした上で、受講希望コースと日程を明記し、ファクシミリでお送りください。

### トレーニング担当

T E L 03-3533-3835  
F A X 03-3533-3781  
E-mail [training@jpn.sas.com](mailto:training@jpn.sas.com)

# Latest Releases

## 最新リリース情報

### PCプラットフォーム

Windows版	SASシステム リリース8.2	TS2M0
OS/2版	SASシステム リリース6.12	TS020
Macintosh版	SASシステム リリース6.12	TS040

### ミニコンピュータプラットフォーム

OpenVMS AXP版	SASシステム リリース6.12	TS020
OpenVMS VAX版	SASシステム リリース6.08	TS407

### UNIXプラットフォーム

MIPS ABI版	SASシステム リリース6.11	TS040
Tru64版	SASシステム リリース8.2	TS2M0
ABI+版	SASシステム リリース6.11	TS040
SunOS/Solaris版	SASシステム リリース8.2	TS2M0
HP-UX版	SASシステム リリース8.2	TS2M0
AIX版	SASシステム リリース8.2	TS2M0
Linux版	SASシステム リリース8.2	TS2M0

### メインフレームプラットフォーム

MVS版	SASシステム リリース6.09E	TS470
MSP版	SASシステム リリース6.09E	TS470
VOS3版	SASシステム リリース6.09E	TS470
CMS版	SASシステム リリース6.08	TS410

# Programmers Workshop

## SASプログラマーズ・ワークショップ開催のお知らせ

SASシステムは、スタンドアローン環境およびクライアント・サーバ環境において、情報系から基幹系までさまざまな利用形態を実現しています。特にオープン系によるクライアント・サーバ環境におけるデータセキュリティについては、どの企業においてもシステムを構築する上で必須条件となっています。

6月のSASプログラマーズ・ワークショップでは、SASシステムを利用する上で、バージョン8eで提供されているセキュリティ機能をどのようにシステムへ反映させるかについて、ご紹介します。

### 東京開催

日程：6月12日(水)  
 会場：株式会社SASインスティテュートジャパン 東京本社  
 住所：東京都中央区勝どき1-13-1 イヌイビル・カチドキ8F  
 プレゼンテーションルーム  
 時間：15:00 ~ 17:00  
 定員：50名

### 大阪開催

日程：6月21日(金)  
 会場：株式会社SASインスティテュートジャパン 大阪支店  
 住所：大阪市北区堂島浜1-4-16 アクア堂島西館12F  
 プレゼンテーションルーム  
 時間：15:00 ~ 17:00  
 定員：50名

両日とも参加は無料となりますので、ぜひご参加ください。なお、定員に限りがございますので、お早めにお申し込みください。お申し込み方法は、「SASプログラマーズ・ワークショップ参加申込用紙」へご記入の上、FAX、E-mailにてご送付ください。SASプログラマーズ・ワークショップのお申し込み用紙は下記のURLより入手できます。

<http://www.sas.com/japan/service/technical/pws.html>

その他、ご不明点、ご質問がございましたら、以下までご連絡ください。

### SASプログラマーズ・ワークショップ担当

T E L 03-3533-3835  
 F A X 03-3533-3781  
 E-mail pws@jpn.sas.com

## SAS Technical News Spring 2002

発行  
株式会社SASインスティテュートジャパン

テクニカルニュースに関するお問い合わせ先

テクニカルサポートグループ

TEL: 03-3533-3877

FAX: 03-3533-3781

E-mail: technews@jpn.sas.com



株式会社SASインスティテュートジャパン

[www.sas.com/japan/](http://www.sas.com/japan/)

東京本社  
〒104-0054  
東京都中央区勝どき1-13-1  
イヌイビル・カチドキ 8F  
Tel 03 (3533) 6921  
Fax 03 (3533) 6927

大阪支店  
〒530-0004  
大阪市北区堂島浜1-4-16  
アクア堂島西館 12F  
Tel 06 (6345) 5700  
Fax 06 (6345) 5655